

CS224 – Prolog search space

Purpose: Prolog searches can be used to solve complex computations such as constraint satisfaction problems.

Knowledge: This activity will help you become familiar with the following content knowledge:

- How to solve constraint satisfaction problems with Prolog

Activity: With your group perform the following tasks and answer the questions. You will be reporting your answers back to the class in 30 minutes.

1. In swipl try the following to investigate Prolog terms. You will see that terms in Prolog are not automatically evaluated:
 - (a) `3 = 1 + 2`. Hmm... why do we get this result?
 - (b) `3 is 1 + 2`. It looks like `is` forces the numeric evaluation.
 - (c) `Y=X+2, X=1`. Why do we get that result?
 - (d) `Y is X+2, X=1`. What is happening and why?
 - (e) `X=1, Y is X+2`. Why is this result different than the one above?
2. A well known constraint satisfaction problem is the eight queens problem to place 8 queens on an empty chess board so that no queen is in check. A queen is in check if it is in the same row, column, or diagonal with another queen. We will just use a term of the form `X/Y` to represent a queen at position `(X,Y)` on the board.

Look at the code for `nocheck` and in your own words explain why this determines that the queen described by `X/Y` is not in check with the queens in the given list:

```
nocheck(_, []).
nocheck(X/Y, [X1/Y1 | Rest]) :-
    X =\= X1,
    Y =\= Y1,
    abs(Y1-Y) =\= abs(X1-X),
    nocheck(X/Y, Rest).
```

Hint: `=\=` means not equal.

3. Look at the code for `legal` which determines if the list is a legal placement of the queens. In your own words, explain what each step in the check is doing:

```
legal([]).
legal([X/Y | Rest]) :-
    legal(Rest),
    member(X, [1,2,3,4,5,6,7,8]),
    member(Y, [1,2,3,4,5,6,7,8]),
    nocheck(X/Y, Rest).
```

4. This code can be improved by observing that we need exactly one queen in each row. Consider the following code which sets this up:

```
/*
    eightqueens(X) succeeds if X is a legal placement of eight queens,
    listed in order of their X coordinates.
*/
eightqueens(X) :-
    X = [1/_,2/_,3/_,4/_,5/_,6/_,7/_,8/_],
    legal(X).
```

There is a line in `nocheck` and a line in `legal` which can now be removed. Which ones?

5. You can copy `~jill/cs224/eightqueens.pl` and do a couple of experiments. Explain what happens and why for each of the changes.
- Change `member(Y, [1,2,3,4,5,6,7,8])` in `legal` to `1=<Y, Y=<8`. On first blush this seems like it would have the same effect. Why not? Change it back.
 - In `nocheck` move `legal(Rest)` from the first item to the last item in the clause. Why is the original order important?